

# Advanced Scheme Techniques

## Some Naughty Bits

Jeremy Brown

January 12 & 14, 2004

## Acknowledgements

Jonathan Bachrach, Alan Bawden, Chris Hanson, Neel Krishnaswami, and Greg Sullivan offered many helpful suggestions on an earlier version of this course.

These slides draw on works by

Hal Abelson, Alan Bawden, Chris Hanson, Paul Graham, Oleg Kiselyov, Neel Krishnaswami, Al Petrofsky, Jonathan Rees, Dorai Sitaram, Gerry Sussman, Julie Sussman, and the R5RS authors group

Thanks also to Scheme Boston, the Boston-area Scheme User's Group.

And of course to SIPB, for organizing.

All errors are, of course, my fault alone.

**Teaser**  
**Using Macros and Continuations Together**

## Backtracking: a Teaser

The “amb” operator always picks an acceptable value:

```
(let ((value (amb 0 1 2 3 4 5 6)))  
  (assert (> value 2))  
  (assert (even? value))  
  value)  
==>
```

## Backtracking: a Teaser

The “amb” operator always picks an acceptable value:

```
(let ((value (amb 0 1 2 3 4 5 6)))  
  (assert (> value 2))  
  (assert (even? value))  
  value)  
==> 4
```

## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==>
```

## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==> (2 1 1)
```

## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==> (2 1 1)
(next)             ==>
```



## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==> (2 1 1)
(next)              ==> (1 2 1)
```

## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==> (2 1 1)
(next)             ==> (1 2 1)
(next)             ==>
```

## Backtracking: a Teaser

```
(define (three-dice sumto)
  (let ((die1 (amb 1 2 3 4 5 6))
        (die2 (amb 1 2 3 4 5 6))
        (die3 (amb 1 2 3 4 5 6)))
    (assert (= sumto (+ die1 die2 die3)))
    (list die1 die2 die3)))

(initialize-amb-fail)
(three-dice 4)      ==> (2 1 1)
(next)             ==> (1 2 1)
(next)             ==> (1 1 2)
```

# **Scheme Macros**

## What is a Macro?

A macro is

- a stylized code transformation...
- performed without evaluating code...
- and using no runtime information.

## What is a Macro?

A macro is

- a stylized code transformation...
- performed without evaluating code...
- and using no runtime information.

Suppose `cond-set!` is a macro:

```
(cond-set! (> test 4) var 15)
```

## What is a Macro?

A macro is

- a stylized code transformation...
- performed without evaluating code...
- and using no runtime information.

Suppose `cond-set!` is a macro:

```
(cond-set! (> test 4) var 15)
```

This expression might expand to:

```
(if (> test 4) (set! var 15))
```

## What is a Macro?

A macro is

- a stylized code transformation...
- performed without evaluating code...
- and using no runtime information.

Suppose `cond-set!` is a macro:

```
(cond-set! (> test 4) var 15)
```

This expression might expand to:

```
(if (> test 4) (set! var 15))
```

**Note: the expansion process does not evaluate `test` or `var`!**



## Macro Expansion Overview

A brief, somewhat inaccurate view of the macro expansion process:

- Pattern-matcher discovers an invocation form with macro keyword in operator position, e.g.  

```
(unless (procedure? f) (display f))
```
- Keyword is associated with one or more pattern/template pairs  
E.g.  

```
<(when condition consequent) ,  
  (if (not condition) consequent)>
```
- If form matches a pattern, the corresponding template is filled in and replaces the form.

## When to Use Macros

Use macros to vary the order of evaluation  
(in other words, to create new syntax/special forms).

For example:

- conditional evaluation (cond, case)
- repeated evaluation (do, named-let)
- binding (let, let\*)
- un-evaluated syntactic tokens (case's =>)

## When NOT to Use Macros

Any time you can avoid it!

- Don't use them for efficiency hacks.

Let the compiler handle that.

Why not?

- Macros aren't first-class objects
  - You can't use a macro as any sort of runtime value
  - Thus, you reduce your development flexibility.
- You make debugging more difficult

## Another Look at cond-set!

Remember our example:

```
(cond-set! (> test 4) var 15)
```

What's wrong with making cond-set! a function? E.g:

```
(define (cond-set! test variable value)  
  (if test (set! variable value)))
```

## Another Look at cond-set!

Remember our example:

```
(cond-set! (> test 4) var 15)
```

What's wrong with making cond-set! a function? E.g:

```
(define (cond-set! test variable value)
  (if test (set! variable value)))
```

The set! only mutates the parameter in the function.

The original var is unchanged.

## Scheme's Derived Special Forms

Scheme actually has very few primitive special forms:

- lambda
- if
- quote
- set!

All the other forms *may* be derived using macros:

- conditionals (cond, case), binding (let, let\*), etc.
- sequencing (begin, and, or)
- iteration (do, named let)

Of course, they may be implemented directly by the compiler, too.

## Scheme Macro Systems

A number of macro systems have been implemented for various Schemes:

- Common Lisp-style *defmacro*
- *syntax-table*
- syntactic closures
- *syntax-case*
- *syntax-rules*
- ...and more!

*syntax-rules* is the macro system endorsed by the “Revised<sup>5</sup> Report on the Algorithmic Language Scheme” (R5RS). *syntax-rules* macros are often called “hygienic” macros.

## Other Macro Systems You May Have Met

- m4
- tex/latex
- cpp



## Key Commonalities Between Lisp-ish Macro Systems

All of these Lisp macro systems share two commonalities.

1. A macro is guaranteed to expand to a valid lisp data structure (typically a list).
2. The expanded form may contain objects that cannot be printed and reparsed.

These are unique: you won't find them in cpp, tex, m4, etc.!

1. helps in generating legal code, since all lisp code is also a valid lisp data structure.
2. helps in avoiding accidental variable capture

# Defmacro

## Defmacro (Common Lisp) I

Defmacro takes

- a name
- an argument list
- a body

Example:

```
(defmacro setq-if-true (lval condn)
  `(let ((tmp ,condn)) (if tmp (setq ,lval tmp))))
```

Terminology:

The `'(let ...)` expression is a *template*.

## Defmacro (Common Lisp) II

At compile time, the body is evaluated to return a lisp form.

Example:

```
(defvar foo)
```

```
(setq-if-true foo 17) ==>
```

```
(let ((tmp 17))
```

```
  (if tmp (setq foo 17))))
```

## Defmacro: Free Variable Capture

You can unintentionally capture free variables in the arguments:


```
(defvar foo 0)
(setq-if-true foo 17) ==>
(let ((tmp 17))
  (if tmp (setq foo 17)))
```

```
(defvar tmp 0)
(setq-if-true tmp 17) ==>
(let ((tmp 17))
  (if tmp (setq tmp 17)))
```


## Defmacro: Variable Capture

You can unintentionally capture free variables in the arguments:

```
(defvar foo 0)
(setq-if-true foo 17) ==>
(let ((tmp 17))
  (if tmp (setq foo 17)))
```




```
(defvar tmp 0)
(setq-if-true tmp 17) ==>
(let ((tmp 17))
  (if tmp (setq tmp 17)))
```




## Defmacro: Variable Capture

You can unintentionally capture free variables in the arguments:

```
(defvar foo 0)
(setq-if-true foo 17) ==>
(let ((tmp 17))
  (if tmp (setq foo 17)))
```



```
(defvar tmp 0)
(setq-if-true tmp 17) ==>
(let ((tmp 17))
  (if tmp (setq tmp 17)))
```



*let* changes a binding, thus introducing a new *syntactic environment*.

(See “Syntactic Closures”, by Alan Bawden and Jonathan Rees, 1988)

## Syntactic Environments

A syntactic environment maps names to meanings

- identifiers map to variables
- keywords map to special syntactic constructs (*lambda*, macros, etc.)



## Syntactic Environments

A syntactic environment maps names to meanings

- identifiers map to variables
- keywords map to special syntactic constructs (*lambda*, macros, etc.)

Binding constructs (*lambda*, *let*, *defmacro*, etc.) introduce new syntactic environments.

## Syntactic Environments

A syntactic environment maps names to meanings

- identifiers map to variables
- keywords map to special syntactic constructs (*lambda*, macros, etc.)

Binding constructs (*lambda*, *let*, *defmacro*, etc.) introduce new syntactic environments.

What code means depends on the syntactic environment in which it is evaluated!

## Counting Syntactic Environments

```
(defmacro setq-if-true (lval condn)
  `(let ((tmp ,condn)) (if tmp (setq ,lval tmp))))
```

```
(defvar tmp 0)
(setq-if-true tmp 17) ==>
(let ((tmp 17))
  (if tmp (setq tmp 17)))
```

There are several syntactic environments:

1. Environments before and after the defmacro
2. Environment of (setq-if-true...)
3. Environment of (let...) in expansion

We want different occurrences of 'tmp' evaluated in different environments!

## Defmacro: *gensym* I

Wise defmacro users extend the syntactic environment with guaranteed-unique names (symbols):

```
(defmacro setq-if-true (lval condn)
  (let ((tmp (gensym)))
    `(let ((,tmp ,condn)) (if ,tmp (setq ,lval ,tmp))))))
```

*gensym* creates a symbol at compile-time, but makes no string-to-symbol mapping. Thus:

- The symbol cannot be found by looking up a string name
- The symbol is thus a unique identifier across all syntactic environments

## Defmacro: *gensym* II

```
(defvar tmp 0)
(setq-if-true tmp 17) ==>
(let ((#:G2762 99))
  (if #:G2762 (setq tmp #:G2762)))
```

Note: This macro expansion cannot be printed and re-parsed:  
a parsed version would not work with code that deliberately named  
a symbol #:G2762

## Defmacro: the Limits of *gensym*

*gensym* can't protect free variables in the macro template.

```
(defmacro push-if (condn expr)
  `(if ,condn (setf stack (cons ,expr stack))))
```

```
(defvar stack '())
(push-if (> 5 3) 7) ==>
  (if (> 5 3) (setf stack (cons 7 stack)))
```


```
(let ((stack '(alpha beta)))
  (push-if (> 5 3) 9)) ==> ; fudging here
(let ((stack '(alpha beta)))
  (if (> 5 3) (setf stack (cons 9 stack))))
```

## Defmacro: the Limits of *gensym*


*gensym* can't protect free variables in the macro template.

```
(defmacro push-if (condn expr)
  `(if ,condn (setf stack (cons ,expr stack))))
```

```
(defvar stack '())
(push-if (> 5 3) 7) ==>
  (if (> 5 3) (setf stack (cons 7 stack)))
```



```
(let ((stack '(alpha beta)))
  (push-if (> 5 3) 9)) ==> ; fudging here
(let ((stack '(alpha beta)))
  (if (> 5 3) (setf stack (cons 9 stack))))
```



# Syntactic Closures



## Syntactic Closures

A *syntactic closure* has three components:

- A syntactic environment
- An expression
- A list of names (symbols and/or syntactic closures over symbols)

## Syntactic Closures

A *syntactic closure* has three components:

- A syntactic environment
- An expression
- A list of names (symbols and/or syntactic closures over symbols)

A syntactic closure represents code in which

- all names in the expression are interpreted according to the syntactic environment...
- *except* names in the list-of-names, which are left free.

## Syntactic Closures

A *syntactic closure* has three components:

- A syntactic environment
- An expression
- A list of names (symbols and/or syntactic closures over symbols)

A syntactic closure represents code in which

- all names in the expression are interpreted according to the syntactic environment...
- *except* names in the list-of-names, which are left free.

*Macros built with syntactic closures have much more control.*

## set-if-true! with MIT Scheme Syntactic Closures

```
(define-syntax set-if-true!  
  (sc-macro-transformer  
    (lambda (exp usage-env)  
      (let ((lval (cadr exp))  
            (condn (caddr exp)))  
        (let ((lval-sc  
              (make-syntactic-closure usage-env '() lval))  
              (condn-sc  
                (make-syntactic-closure usage-env '() condn)))  
          `(let ((tmp ,condn-sc))  
              (if tmp (set! ,lval-sc tmp))))))))))
```

## Syntactic Closures: Expansion

These both work safely now:

```
(set-if-true! tmp 17) ==>
(let ((tmp 17))
  (if tmp (set! tmp tmp)))
```

```
(let ((let 7)) (set-if-true! foo 17)) ==> ; fudging
(let ((let 7))
  (let ((tmp 17))
    (if tmp (set! tmp tmp))))
```

## Syntactic Closures: Expansion

These both work safely now:

```
(set-if-true! tmp 17) ==>
(let ((tmp 17))
  (if tmp (set! tmp tmp)))
```

```
(let ((let 7)) (set-if-true! foo 17)) ==> ; fudging
(let ((let 7))
  (let ((tmp 17))
    (if tmp (set! tmp tmp))))
```

Post-expansion, *tmp* and *let* are not symbols.  
Instead, they are syntactic closures.

## Deliberately Capturing Variables

```
(define-syntax this-lambda
  (sc-macro-transformer
    (lambda (exp usage-env)
      (let ((lambda-sc
            (make-syntactic-closure usage-env
              '(this)
              `(lambda ,@(cdr exp))))))
        `(letrec ((this ,lambda-sc))
           this))))))
```

### Fibonacci:

```
((this-lambda (n)
  (if (<= n 2) 1 (+ (this (- n 1)) (this (- n 2))))) 7) ==>
13
```

## Take-Home Lessons so Far

- Explicit destructuring of input forms is ugly.
- Explicit management of environments is verbose.
- There are usually three syntactic environments that matter:
  1. Macro definition environment
  2. Macro usage environment
  3. Environments created by the macro expansion, *if* you want to deliberately introduce bindings into the expansion of the macro-user's code.



# **syntax-rules**

## **R5RS Macros**

### **(AKA Hygienic Macros)**

### **(AKA *syntax-rules* macros)**

Scheme “hygienic macros” have four key properties:

- Hygiene: macros don’t interfere with their expansion environments
- Referential transparency: macros aren’t interfered with by their expansion environments
- Pattern-language: specify how to destructure input forms
- Closed: the macro language is decoupled from base Scheme; you cannot call Scheme functions as part of the expansion of a macro

## Macro Hygiene

*Hygienic* macros cannot contaminate the lexical scope in which they expand by introducing symbols that shadow other bindings.

## Macro Hygiene

*Hygienic* macros cannot contaminate the lexical scope in which they expand by introducing symbols that shadow other bindings.

Equivalent statement:

Hygienic macros close their input forms in the syntactic environment at the point of use.

## Referential Transparency

*Referentially-transparent* macros are not contaminated by the lexical scope in which they expand.

Free identifiers introduced by the template refer to their bindings at the point of macro definition.

## Referential Transparency

*Referentially-transparent* macros are not contaminated by the lexical scope in which they expand.

Free identifiers introduced by the template refer to their bindings at the point of macro definition.

Equivalent statement:

A referentially-transparent macro closes identifiers in its template in the syntactic environment at the point of definition.

## Pattern Language

A basic R5RS macro pattern is pretty straightforward:

- it is a list form
- its first element is the keyword
- strings, numbers, booleans, lists, vectors represent themselves
- non-keyword symbols represent *pattern variables*

For a form to match a pattern:

- each number, boolean, etc. must match exactly
- each pattern variable matches a single subform

Unaddressed so far: how do we represent specific symbols?

We'll come back to that in awhile.

## Pattern Language Examples I

### Pattern

```
(let1 (name value) body)
```

### matches form

```
(let1 (x (read))  
      (if (not x) (display "you said no")))
```

with the pattern variables matching like this:

```
name = x  
value = (read)  
body = (if (not x) (display "you said no"))
```



## Pattern Language Examples II

Pattern

```
(contrived #((first . rest) #(3 any)))
```

matches form

```
(contrived #((1 2 3 4 5) #(3 '(foo))))
```

with the pattern variables matching like this:

```
first = 1
```

```
rest = (2 3 4 5)
```

```
any = '(foo)      AKA  (quote (foo))
```

## Template Language

A template is an arbitrary Scheme form whose interpretation depends on the pattern it's paired with.

- numbers, booleans, lists, vectors represent themselves
- symbols which don't appear in the pattern represent themselves
- symbols which do appear in the pattern represent pattern variables

Expansion replaces each pattern variable in a template with the subform it matched in the input form.

## Template Language Example

### Pattern

```
(let1 (name value) body)
```

### and template

```
(let ((name value)) body)
```

### applied to form

```
(let1 (x (read))  
  (if (not x) (display "you said no")))
```

### expands to

```
(let ((x (read)))  
  (if (not x) (display "you said no")))
```

## Matching Multiple Forms at Once

A pattern variable followed by ... (an ellipsis) matches a group of consecutive forms.

For example, if we match the pattern

```
(dotimes count statement ...)
```

against the code form

```
(dotimes 5 (set! x (+ x 1)) (display x))
```

then

```
statement ... = (set! x (+ x 1)) (display x))
```

## Template Expansion with Ellipses I

In a template, a pattern variable followed by an ellipsis expands into the group of forms it matched.

E.g. given this template for `dotimes`

```
(let dotimes-loop ((counter count))
  (if (> counter 0)
      (begin
        statement ...
        (dotimes-loop (- counter 1))))))
```

## Template Expansion with Ellipses II

...then the expansion will look like this:

```
(let dotimes-loop ((counter 5))
  (if (> counter 0)
      (begin
        (set! x (+ x 1))
        (display x)
        (dotimes-loop (- counter 1))))))
```

## Producing Repeated Forms

Suppose we want

```
(thunkify 5 (* x x))
```

to expand to

```
(list (lambda () 5) (lambda () (* x x)))
```

This does the trick:

Pattern: (thunkify body ...)

Template: (list (lambda () body) ...)

## Matching Repeated Forms

Suppose we want

```
(update-if-true!  
  (> x 5) x-is-big)  
  ((zero? y) y-is-zero))
```

to expand to

```
(begin  
  (let ((test (> x 5)))  
    (if test (set! x-is-big test)))  
  (let ((test (zero? y)))  
    (if test (set! y-is-zero test))))
```



## Matching Repeated Forms II

We can match a group of forms by following a form with ...  
pattern variables in the form match the corresponding subforms.

This does the trick:

Pattern:

```
(update-if-true! (condition variable) ...)
```

Template:

```
(begin (let ((test condition))  
  (if test (set! variable test))) ...)
```

## Nesting Ellipses

Ellipses may be nested in both patterns and templates.

A highly artificial example: we want this

```
(quoted-append (1 2 3) (a b c) (+ x y))
```

to expand into this

```
'(1 2 3 a b c + x y)
```

This does it:

```
Pattern: (quoted-append (guts ...) ...)
```

```
Template: (quote (guts ... ...))
```

**This can be tricky!**

## Grouping Pattern/Template Pairs

A keyword may be associated with multiple pattern/template pairs. The complete ruleset for a keyword is given by a *syntax-rules* form, for instance this syntax-rules for *and* from R5RS:

```
(syntax-rules ()
  ((and) #t) ; first pair
  ((and test) test) ; second pair
  ((and test1 test2 ...) ; third pattern
   (if test1 (and test2 ...) #f))) ; third templ.
```

In MIT-Scheme terms, *syntax-rules* returns a *macro transformer*.

## Notes on syntax-rules

- Patterns may contain their keyword, causing recursive expansion!
- Forms are matched against patterns in top-down order
- and *syntax-rules* solves another problem for us...

## Representing Specific Symbols in Patterns

Suppose we want

```
(implications (a => b) (c =>d) (e =>f))
```

to expand to

```
(begin (if a b) (if c d) (if e f))
```

But " $=>$ " is a scheme symbol just like "foo"; if we write

```
(syntax-rules ()
  ((implications (condition => consequent) ...)
   (begin (if condition consequent) ...)))
```

```
(implications (test =. (set! testp #t))) ;typo
```

then " $=>$ " would match `=.`, instead of the expander signaling an error.

## Representing Specific Symbols in Patterns II

We can specify a list of non-pattern-variable symbols as part of syntax rules, for example

```
(syntax-rules (=>)  
  ((implications (condition => consequent) ...)   
   (begin (if condition consequent) ...)))
```

Now the “=>” in the pattern will only match the symbol “=>”; it is no longer a pattern variable.

## Binding Constructs

We are now ready to use syntax-rules for real. But how?

We have three options:

- `(let-syntax bindings body)`
- `(letrec-syntax bindings body)`
- `(define-syntax symbol (syntax-rules ....`  
`(top level only)`

All behave in an extraordinarily obvious way.

## Representing Specific Symbols in Patterns III

One final subtlety arises. Consider

```
(define-syntax implications
  (syntax-rules (=>) ; body elided
  ))
```

```
(let ((=> 5))
  (implications (foo => bar)))
```

In the define-syntax form, “=>” names an implicit top-level binding.

In the implications form, “=>” names the let binding.

Because of this, they do not match. Thus,

**in this lexical context, the expansion of implications will fail.**



## Representing Specific Symbols in Patterns IV (Final)

In general:

- A literal symbol in a syntax-rules names a binding in the lexical scope of the syntax-rules.
- A symbol in a form names a binding in the lexical scope in which the form appears.
- A symbol in a form will only match a literal symbol in a pattern if both symbols name the same binding.
- (A symbol which doesn't correspond to an explicit binding is assumed to correspond to an implicit binding in the top level.)

## Debugging Tips

- Quote the result of an expansion that's giving you trouble in order to see the intermediate result.
- If MIT scheme says “Hardware trap SIGSEGV” when you define a macro, it means you have ellipses that don't make sense.
- MIT scheme doesn't like ellipses in vectors, i.e. `#( a . . . )` buys you a SIGSEGV message.

## **R5RS Macro Semantics**

## Macro Expansion: Apparent Danger

```
(define-syntax dotimes
  (syntax-rules ()
    ((dotimes count body ...)
     (let loop ((counter count))
       (if (> counter 0)
           (begin
              body ...
              (loop (- counter 1))))))))

(define counter 0)
(dotimes 5 (set! counter (+ counter 1)))
```

Does the dotimes invocation ever terminate?

## Hygienic Dotimes Expansion

A hygienic expansion contains the usual extra information:

```
(let loop ((counter 5))
  (if (> counter 0)
      (begin
        (set! counter (+ counter 1))
        (loop (- counter 1))))))
```

**counter** is not the same as counter

It was closed in a different syntactic environment. It is therefore not captured by the let.

## Substitutions *Can* Shadow

Using pattern-variables, macros can create bindings that shadow lexically-enclosing bindings.

```
(define-syntax shadow
  (syntax-rules () ((shadow used-arg body)
                    (let ((used-arg 5)) body))))

(define test 7)
(shadow test test)
```

yields

```
(let ((test 5)) test) ==> 5
```

This technique is how binding constructs are implemented.

## Substitutions *Can* Shadow

Using pattern-variables, macros can create bindings that shadow lexically-enclosing bindings.

```
(define-syntax shadow
  (syntax-rules () ((shadow used-arg body)
                    (let ((used-arg 5)) body))))

(define test 7)
(shadow test test)
```

yields

```
(let ((test 5)) test) ==> 5
```

This technique is how binding constructs are implemented.

Syntactic closure over a binding form does the right thing!

## Another Apparent Danger

```
(define-syntax dotimes
  (syntax-rules ()
    ((dotimes count body ...)
     (let loop ((counter count))
       (if (> counter 0)
           (begin
              body ...
              (loop (- counter 1))))))))
```

```
(let ((- 'minus))
  (dotimes 5 (display "hello, world"))
```

Does `(- counter 1)` evaluate properly?



## Referentially Transparent Dotimes Expansion

```
(let ((- 'minus))
  (let loop ((counter 5))
    (if (> counter 0)
        (begin
          (display "hello, world")
          (loop (- counter 1))))))
```

- is not the same as -; they are closed in different syntactic environments.

## R5RS Macros are Lexically-Safe Macros

*Hygiene* and *referential transparency* combine to make R5RS macros *lexically safe*: lexical scoping is always preserved. In particular:

- macros don't interfere with their expansion environments (hygiene)
- macros aren't interfered with by their expansion environments (referential transparency)
- temporary names introduced by recursive/nested macros never collide
- "Principle of least surprise"

## An Implication of lexical safety

The implication:

- Hygiene + referential transparency guarantees lexical scoping at all times...
- ...even when you don't want it to. (We'll touch on that later.)

## On Not Being Scheme

The R5RS macro language is decoupled from base Scheme

Most (all?) other macro systems let you use Scheme as part of the macro expansion process, i.e.

- Some Scheme is executed at compile-time, to help produce...
- Scheme which is executed at runtime

Such systems enable you to, e.g, generate new symbols using string->symbol, etc. at compile-time.

## **R5RS Design Idioms**

## Ordering Multiple Patterns

A form is compared against a keyword's patterns in top-down order.

In dealing with things that process lists in order:

- The first form should match zero items
- The second should match one item
- The third should generalize for more than one item

For example, *and...*

## Ordering Multiple Patterns: R5RS Example

```
(define-syntax and
  (syntax-rules ()
    ((and) #t)           ; zero items
    ((and test) test)   ; one item
    ((and test1 test2 ...) ; two items
     (if test1 (and test2 ...) #f))))
```

Your macros may not need all three cases.

## Staged Expansion

Consider this pair of macros:

```
(define-syntax reverse-and-quote-list
  (syntax-rules ()
    ((reverse-and-quote-list list)
     (rl-helper list ())))))
```

```
(define-syntax rl-helper
  (syntax-rules ()
    ((rl-helper () (backw ...))
     '(backw ...))
    ((rl-helper (arg rest ...) (backw ...))
     (rl-helper (rest ...) (arg backw ... ))))))
```

```
(reverse-and-quote-list (1 2 3 4 5)) ==> (5 4 3 2 1)
```

rl-helper can't be nested: it has ... in it



## Staged Expansion Using Keystings

Use “keystings” to stage expansion without cluttering the namespace with innumerable helper macros.

```
(define-syntax reverse-and-quote-list
  (syntax-rules ()
    ((reverse-and-quote-list "helper" () (backw ...))
     '(backw ...))
    ((reverse-and-quote-list "helper" (arg rest ...)
                              (backw ...))
     (reverse-and-quote-list "helper" (rest ...)
                              (arg backw ... )))
    ((reverse-and-quote-list (list ...))
     (reverse-and-quote-list "helper" (list ...) ())))))
```

“Tail recursive” strategy.

## Macro Subroutines: “Falling-Forward”

Consider:

```
(let ((letrec 5)) letrec) ==> 5
```

The expansion process can reorganize the inner forms, so the outermost layer of a form is always expanded first.

The next expansion doesn't begin until the first one is completely done.

How can an macro call another macro as a subroutine, and continue expanding afterward?

## Macro Subroutines

### Passing the Next Macro as an Argument

To use a macro as a subroutine, we pass along the “return” as pair of arguments to be used in a “tail-call” strategy.

- *future-keyword* names the macro to be applied when this one is through.
- *future-args* provides initial arguments to that macro.

Example:

```
(define-syntax cps-quote
  (syntax-rules ()
    ((cps-quote future-keyword (future-args ...) stuff ...)
     (future-keyword future-args ... (quote stuff ...))))))
```

## Terminal Macros

Some macros are “terminal” — they don’t call any more macros:

```
(define-syntax apply-to-result
  (syntax-rules ()
    ((apply-to-result func list ...)
     (func list ...))))
```

We can use this to output a quoted value:

```
(cps-quote apply-to-result ((lambda (x) x)) (1 2 3 4 5))
==>
(1 2 3 4 5)
```

## Staged Subroutine Macros

Staged macros pass their future-arguments along until they're done:

```
(define-syntax cps-reverse-list
  (syntax-rules ()
    ((cps-reverse-list "helper" future-keyword
      (future-args ...) () (backw ...))
     (future-keyword future-args ... (backw ...)))
    ((cps-reverse-list "helper" future-keyword
      future-args (arg forw ...) (backw ...))
     (cps-reverse-list "helper" future-keyword
      future-args (forw ...) (arg backw ...)))
    ((cps-reverse-list future-keyword future-args (list ...))
     (cps-reverse-list "helper" future-keyword future-args
      (list ...) ())))))
```

## Nesting

The more “subroutines” you want to call, the longer the chain:

```
(cps-reverse-list
  cps-quote
  (apply-to-result ((lambda (x) x))) (1 2 3 4 5))
==>
  (5 4 3 2 1)
```

## **CPS (Continuation Passing Style)**

At any point in a macro expansion series, the future-keyword/future-args pair represent all of the expansion to come.

We say they represent the “continuation” of the macro expansion.

The subroutining style is called Continuation Passing Style (CPS).

But we’ve only talked about macro-continuations, and macros aren’t first-class.

**Tomorrow:**  
**First Class Scheme Continuations**

They ain't macros...



**Tomorrow:  
First Class Scheme Continuations**

They ain't macros...

but first...

## **R5RS Macro Limitations...**

## **R5RS Macro Limitations...**

...and Workarounds

## Hygiene-Imposed Limitation

Suppose we *want* `dotimes` to expose a variable counter, so that a programmer could write

```
(dotimes 5 (display counter))
```

and get “54321” as the result.

We already know that hygiene prevents this definition from working that way:

```
(define-syntax dotimes
  (syntax-rules ()
    ((dotimes count body ...)
     (let loop ((counter count))
       ....
```

## The Workaround

We know that we *can* use symbols from the arguments to the macro to create bindings, e.g. `let`, `letrec`, etc.

In fact, if the arguments contain the identifier you're looking for (e.g. `counter`) anywhere, you can dig it out and use it!

## Petrofsky's find-identifier

```
(define-syntax find-identifier
  (syntax-rules ()
    ((_ ident (x . y) sk fk)
     (find-identifier ident x sk
                      (find-identifier ident y sk fk)))
    ((_ ident #(x ...) sk fk)
     (find-identifier ident (x ...) sk fk))
    ((_ ident form sk fk)
     (let-syntax
       ((check
         (syntax-rules (ident)
           ((_ ident ident* (s-f . s-args) fk_)
            (s-f ident* . s-args))
           ((_ x y sk_ fk_) fk_))))
      (check form form sk fk))))))
```

## “Hygiene-Violating” dotimes

```
(define-syntax dotimes
  (syntax-rules ()
    ((dotimes count body ...)
     (find-identifier counter (body ...)
                        (dotimes-finish count body ...)
                        (dotimes-finish temp count body ...))))))
```

```
(define-syntax dotimes-finish
  (syntax-rules ()
    ((dotimes-finish counter count body ...)
     (let loop ((counter count))
       (if (> counter 0)
           (begin
              body ...
              (loop (- counter 1))))))))))
```

## And Sure Enough

```
(dotimes 5 (display counter))  
prints 54321
```



**C++ Joke:**  
**Are We Template-Metaprogramming Yet?**

## Homework Problem: Nesting Scopes

We expect this

```
(dotimes 5 (dotimes 5 (display counter)))
```

to yield

```
5432154321543215432154321
```

But actually we get

```
5555544444333332222211111
```

It can be solved...

...but it is tricky.

## **The Other Commonly-Complained About R5RS Limitation**

You cannot synthesize symbols, and thus you cannot make a macro  
(`define-structure foo`) which defines `make-foo`,  
`is-foo?`, etc.

The workaround I've seen is ugly or unportable.

**The End**

But come back tomorrow for the Continuations lecture!